
	Analyzing data – tips, tricks

and experiences

Session 4 – creating graphs


	General
	This session is used for the basic ideas in creating graphs.

	Some examples
	Run the following macros to see different entities, all created using the graphing abilities:

%Taxi, %Birthday, %DoE1 (fig 2), %DoE2, %LinC (incl drawing), %Coord, %Chain (both pages), %Geopdf (all pages), %Hypo, %Stratify


 Creating a graph (similar to the graph of %Stratify, fig 1)
Before starting the creation of the graph the data is created. There is one Y-variable (c1 named 'Result') and one X-variable (c2 named 'X'). The data consists of three groups with a group indicator stored in c5. When doing so-called regression analysis column c5 can not be used as is. Even if c5 is numeric is must be considered as labels only. Therefore c5 is coded by c3 and c4 (see more in "C:\Program Files\MINITAB 15\Macros\Exercises\DOE, regression\About Regression.doc" and the macro %Reg).

The Y-variable consists of a model plus a random part, see below:
name c1 'Result' c2 'X' c3 'Z1' c4 'Z2' c5 'Grp' c7 'Noise'

random 51 c2;                         # Creates X-data, uniformly distri-

uniform 0 9.                          # buted of the range 0 – 9.

set c3                                # C3 and c4 together performs the
(0 1 0)17                             # coding of the three groups.

end                                   #

set c4                                #

(0 0 1)17                             #

end                                   #

set c5                                # Gives the group numbers (1, 2, 3)

(1 2 3)17

end

random 51 c7;                         # The random noise to be added to the model.

normal 0 2.

let c1 = 17.5 + 2.1 * c2 + 4.2 * c3 + 6.6 * c4 + c7   # Model plus random part.
The structure of the plot command.  The basic plot command is fairly simple:

plot c1 * c2                          # Makes a diagram of Y against X.

Substructures of the plot command.  There is a large number of subcommands that are used to add lines, describe plot symbols, position texts, etc, etc. These subcommands have in their turn subcommands. The order of subcommands below a PLOT-command is not important but each next subcommand must 'belong to its mother'. An example: 

plot c1*c2;                           # The 'plot'-command (the top level).
.

symbol;                               # Symbol command and its substructure.
  type 6;                             # These four lines must be together and
  color 4;                            # there is no particular order of 'type',
  size 3;                             # 'color' and 'size'. (Indents are not needed).

text k1 k2 'A short text';            # A 'text'-structure at XY specified by k1 and k2

  tsize 0.6;                          # text size 0.6 and text colour 3 (green).

  tcolor 3;

.
A comment to the DATA-command.  Below the DATA-command is used for positioning the plot in the graph window. The area of the graph window has the size 1 where the lower left corner is designated X=0, Y=0 and upper right corner is designated as X=1, Y=1 (thus the area is 1). The Data command DATA 0.08 0.68 0.18 0.83 specifies that the plot will be within a rectangle with its left vertical limit 0.08 from the left edge and its right vertical limit 0.68 from the left edge of the graph window. 

Likewise the plot is above the rectangle's lower limit which is 0.18 from the bottom edge of the graph window, finally the plot is below the upper limit of the rectangle at 0.83. (The easiest way to understand this is to repeatedly change the DATA-command below remembering that the four values must be within the range 0-1 and usually two decimal places are enough to accurately place the plot where desired.
The following commands will create a graph. It is possible to change the appearance by omitting rows or com​mands. However, some care must be used when removing lines not to confuse the structure: 

plot c1*c2;                           #
grid 1;                               # Grid lines starting from the X-axis (1st axis).
grid 2;                               # Grid lines starting from the Y-axis (2nd axis).
Scale 1;                              # Certain features for the X-axis:
  tsize 0.6;                          # Text size 0.6 for the tick labels.
  HDisplay 0 0 0 0;                   # Removes the 'upper' X-scale. 
Scale 2;                              # The scale info for the Y-axis.
  tsize 0.6;                          #
  HDisplay 0 0 0 0;                   #
nodtitle;                             # No default title of the graph.
symbol;                               # Commands for describing type of symbol,
  type 6;                             # colour and size. NB that the colour for each dot
  color c5;                           # is stored in c5 (the 'group variable').
  size 0.7;                           #
data 0.08 0.68 0.18 0.83;             # Specifies the plot area within the graph window.
  type 0;                             # Specifies that the plot area is transparent.
  etype 0;                            # Gives no edge around the plot area.
axlabel 1 'Values of the variable X'; # Specifies the label of the X-axis.
  tsize 0.7;                          # Gives the text size.
axlabel 2 '- Result -';               # Specifies the label of the Y-axis.
  tsize 0.7.                          # Gives the text size.

Introduction to the LAYOUT-command.  When e.g. placing several plots in the graph window the LAYOUT-command is very handy (see also further below). It can be used to place titles and other annotations on the graph window. Below both time and some text is placed by the help of a number of commands:

# Getting the date and time to be placed in lower right corner of the graph.
erase c3991                           # Erases a column e.g. c3991

let c3991(1) = now()                  # Places current date/time in c3991

Date c3991 c3991;                     # Gives it proper format.

Format(DT40yyyy-mm-dd hh:mm).

Text c3991 c3991                      # Makes it into TEXT-format.

Let k952 = c3991(1)                   # Storage in constant for the graph.
layout;                               # Opens the LAYOUT (closed by ENDLAYOUT below).

wtitle 'A stratified diagram';        # Title of the window on the PC.

title 'A stratified diagram';         # Title text.

tsize 1.1;                            # Text size.

title 'Black dots: process A.  Red dots: process B.  Green dots: process C';

tsize 0.9;                            # Text size of the title text.

tcolor 4;                             # Text colour of the title text..

text 0.86 0.015 k952;                 # Placing time/date in lower right corner of graph window.

tsize 0.5.                            # Text size.

plot c1*c2;                           #
grid 1;                               # Grid lines starting from the X-axis (1st axis).
grid 2;                               # Grid lines starting from the Y-axis (2nd axis).
Scale 1;                              # Certain features for the X-axis:
  tsize 0.6;                          # Text size 0.6 for the tick labels.
  HDisplay 0 0 0 0;                   # Removes the 'upper' X-scale. 
Scale 2;                              # The scale info for the Y-axis.
  tsize 0.6;                          #
  HDisplay 0 0 0 0;                   #
nodtitle;                             # No default title of the graph.
symbol;                               # Commands for describing type of symbol,
  type 6;                             # colour and size. NB that the colour for each dot
  color c5;                           # is stored in c5 (the 'group variable').
  size 0.7;                           #
data 0.08 0.68 0.18 0.83;             # Specifies the plot area within the graph window.
  type 0;                             # Specifies that the plot area is transparent.
  etype 0;                            # Gives no edge around the plot area.
axlabel 1 'Values of the variable X'; # Specifies the label of the X-axis.
  tsize 0.7;                          # Gives the text size.
axlabel 2 '- Result -';               # Specifies the label of the Y-axis.
  tsize 0.7.                          # Gives the text size.

endlayout                             # Closes the LAYOUT-command.

Adding some text.  Adding text to the graph can be done in several different ways. Below we first specify a number of parameters of e.g. size, colour, position, etc. The text is placed in column c801 and with some addi​tional commands in the LAYOUT the text is placed on the graph window:

# Some text (in c801 below) to be placed on the graph. The following 3 rows

# must be added to the LAYOUT-command further below.

# text c802 c803 c801;                # X- and Y-position of the text (c802-c803) and the text (c801).

#  tsize k801;                        # Constant k801 contains the size of the text.

#  tcolor k802.                       # Constant k802 contains the colour of the text (an integer).

let k801 = 0.6                        # Textsize.

let k802 = 1                          # Text colour.

let k803 = 0.92                       # Position of top row in Y.

let k804 = 0.69                       # Position in X.

let k805 = 0.03                       # Line spacing.

#   k806                                Number of rows.

#   k807                                Y-position of bottom row.

read c801;                            # The text.

format(a80).

'Stratify' means that the data is divided into its

natural components such as 'supplier', 'machine', 

'process', 'design', etc.

By doing this, there is a better chance of

understanding the variation, find differences

between processes, suppliers, etc.

end

let k806 = count(c801)                # Number of rows.

set c802                              # X-coordinates of the text rows.

k806(k804)

end

let k807 = k803 - (k806-1)*k805       # Y-coordinate of last row.

set c803                              # Y-coordinates of the text rows.

k803:k807/k805

end
layout;                               # Opens the LAYOUT (closed by ENDLAYOUT below).

wtitle 'A stratified diagram';        # Title of the window on the PC.
title 'A stratified diagram';         # Title text.
  tsize 1.1;                          # Text size.
title 'Black dots: process A.  Red dots: process B.  Green dots: process C';

  tsize 0.9;                          # Text size.
  tcolor 4;                           # Text colour.
text 0.86 0.015 k952;                 # Placing time/date in lower right corner of graph window.
  tsize 0.5;                          # Text size.

text c802 c803 c801;                  # X- and Y-position of the text (c802-c803) and the text (c801).

  tsize k801;                         # Constant k801 contains the size of the text.

  tcolor k802.                        # Constant k802 contains the colour of the text (an integer).

plot c1*c2;                           #
grid 1;                               # Grid lines starting from the X-axis (1st axis).
grid 2;                               # Grid lines starting from the Y-axis (2nd axis).
Scale 1;                              # Certain features for the X-axis:
  tsize 0.6;                          # Text size 0.6 for the tick labels.
  HDisplay 0 0 0 0;                   # Removes the 'upper' X-scale. 
Scale 2;                              # The scale info for the Y-axis.
  tsize 0.6;                          #
  HDisplay 0 0 0 0;                   #
nodtitle;                             # No default title of the graph.
symbol;                               # Commands for describing type of symbol,
  type 6;                             # colour and size. NB that the colour for each dot
  color c5;                           # is stored in c5 (the 'group variable').
  size 0.7;                           #
data 0.08 0.68 0.18 0.83;             # Specifies the plot area within the graph window.
  type 0;                             # Specifies that the plot area is transparent.
  etype 0;                            # Gives no edge around the plot area.
axlabel 1 'Values of the variable X'; # Specifies the label of the X-axis.
  tsize 0.7;                          # Gives the text size.
axlabel 2 '- Result -';               # Specifies the label of the Y-axis.
  tsize 0.7.                          # Gives the text size.

endlayout                             # Closes the LAYOUT-command.

Adding the three regression lines.  The diagram created by %Stratify (fig 1) contains three lines that are the result from the so-called regression analysis (see the reference above). The mathematical model fitted to the data contains one dependent variable (Y) and two X-variables, 'X' and the 'group' variable where these groups are coded by the two Z-variables (Run the %Stratify for more information). 

The following commands will perform the regression analysis but in this case we only want the coefficients of the fitted linear model:

regression c1 3 c2-c4;                # Performs the regression analysis (type e.g. HELP REGR).
coefficients c9.                      # The coefficients are stored in c9.

name c9 'Regression coeff'            # Names column c9.
# The model:      Result = b0 + b1*X + b2*Z1 + b3*Z2
# NB that b0, b1, b2, b3 are the coefficients stored in c9 and that 

# Z1 and Z2 are either '0 0', '1 0' or '0 1' in the expressions below.

# If a term is zero it is not included (to save space). 

# The 12 constants contain the end points (X,Y) for the three lines.

let k1 = min(c2)                      # Minimum X-value.
let k3 = max(c2)                      # Maximum X-value.
let k2 = c9(1) + c9(2)*k1             # Calculating corresponding Y-value for the min X.

let k4 = c9(1) + c9(2)*k3             # Calculating corresponding Y-value for the max X.
let k6 = c9(1) + c9(2)*k1 + c9(3)     # Calculating corresponding Y-value for the min X.
let k8 = c9(1) + c9(2)*k3 + c9(3)     # Calculating corresponding Y-value for the max X.
let k10 = c9(1) + c9(2)*k1 + c9(4)    # Calculating corresponding Y-value for the min X.
let k12 = c9(1) + c9(2)*k3 + c9(4)    # Calculating corresponding Y-value for the max X.
layout;                               # Opens the LAYOUT (closed by ENDLAYOUT below).

wtitle 'A stratified diagram';        # Title of the window on the PC.
title 'A stratified diagram';         # Title text.
  tsize 1.1;                          # Text size.
title 'Black dots: process A.  Red dots: process B.  Green dots: process C';

  tsize 0.9;                          # Text size.
  tcolor 4;                           # Text colour.
text 0.86 0.015 k952;                 # Placing time/date in lower right corner of graph window.
  tsize 0.5;                          # Text size.

text c802 c803 c801;                  # X- and Y-position of the text (c802-c803) and the text (c801).

  tsize k801;                         # Constant k801 contains the size of the text.
  tcolor k802.                        # Constant k802 contains the colour of the text (an integer).
plot c1*c2;                           #
grid 1;                               # Grid lines starting from the X-axis (1st axis).
grid 2;                               # Grid lines starting from the Y-axis (2nd axis).
Scale 1;                              # Certain features for the X-axis:
  tsize 0.6;                          # Text size 0.6 for the tick labels.
  HDisplay 0 0 0 0;                   # Removes the 'upper' X-scale. 
Scale 2;                              # The scale info for the Y-axis.
  tsize 0.6;                          #
  HDisplay 0 0 0 0;                   #
nodtitle;                             # No default title of the graph.
symbol;                               # Commands for describing type of symbol,
  type 6;                             # colour and size. NB that the colour for each dot
  color c5;                           # is stored in c5 (the 'group variable').
  size 0.7;                           #
data 0.08 0.68 0.18 0.83;             # Specifies the plot area within the graph window.
  type 0;                             # Specifies that the plot area is transparent.
  etype 0;                            # Gives no edge around the plot area.
axlabel 1 'Values of the variable X'; # Specifies the label of the X-axis.
  tsize 0.7;                          # Gives the text size.
axlabel 2 '- Result -';               # Specifies the label of the Y-axis.
  tsize 0.7;                          # Gives the text size.

line k1 k2 k3 k4;                     # A line XY left end, XY right end.

size 2;                               # Width 2.

color 1;                              # Colour black.

line k1 k6 k3 k8;

size 2;

color 2;                              # Colour red.

line k1 k10 k3 k12;

size 2;

color 3.                              # Colour green.

endlayout                             # Closes the LAYOUT-command.

 A programming task: the ‘SEMC’-sun
This exercise consists of using Minitab commands to create a sun with its rays. Below are a few remarks. 

1. The center of the sun is to be placed randomly (uniformly distributed in X and Y) in the diagram

2. The number of rays must follow an integer distribution

3. The rays are to point in a random angle (uniformly distributed over 0 to 6.28 (= 2 radians))

4. The lengths of the rays will follow a normal distribution

5. Use trigonometric functions to recalculate from polar to Cartesian (X, Y) coordinates

6. In the center of the sun the text ”SEMC” will be placed

7. All constants needed must be placed at the top of the macro and be commented

The rays need to be drawn as one long line using the POLYGON-command. 
Let k1 = 20                          # Size of the sun.

Let k2 = 0.10                        # k2 and k3 gives the limiting rectangle for c1 c2

Let k3 = 0.90                        # i.e. the position of the center of the sun.

Let k4 = 7                           # Colour of the sun and the rays.

Let k5 = 20                          # Lower end of number of rays.

Let k6 = 50                          # Upper end of number of rays.

Let k8 = 0.6                         # Mean length of rays.

Let k9 = 0.1                         # Standard deviation of lengths of rays.

Let k10 = 4                          # Width of rays.

name c1 'Y-coord' c2 'X-coord' c3 'Number of rays' c4 'angles' c5 'length of rays'

name c6 'end of ray (X)' c7 'end of ray (Y)'

Random 1 c1 c2;                      # The coordinates (the center) of the sun.

Uniform k2 k3.                       # NB only one row is generated.
Random 1 c3;                         # Number of rays is an integer between

integer k5 k6.                       # k5 and k6.

Let k7 = c3(1)                       # Number of rays stored in k7.

Random k7 c4;                        # The angles of the rays in radians.

Uniform 0 6.28.

Random k7 c5;                        # The lengths of the rays.

Normal k8 k9.                        # Expected value k8, standard deviation k9.

Let c5 = abs(c5)                     # Removes negative values of ray lengths.

Let c6 = c5 * cos(c4)                # Gives X-value of end of each ray.

Let c7 = c5 * sin(c4)                # Gives Y-value of end of each ray.

Let k11 = c2(1)                      # X-coordinate of the sun.

Let k12 = c1(1)                      # Y-coordinate of the sun.

Let c6 = c6 + k11                    # Adds the position of the sun.

Let c7 = c7 + k12                    
# ----------------- The following lines prepare the plotting of the rays.

name c11 'X-coords of sun' c12 'Y-coords of sun' c13 'all X-values' c14 'all Y-values'

name c19 'X of ray-tip' c20 'Y of ray-tip'

Set c11                              # Stores X-coordinate of sun k7 times in c11.

K7(k11)

end

Set c12                              # Stores Y-coordinate of sun k7 times in c12.

K7(k12)

end

stack c6 c11 c13                     # Stacks all X-values.

stack c7 c12 c14                     # Stacks all Y-values.

Set c15                              # Stores the values 1 to k7 in column c15.

1:k7                                 # k7 is number of rays.

end

let c16 = c15 + 0.5                  # Adds 0.5 to c15, stores the result in c16.
stack c15 c16 c17                    # Stacks c15 on c16 and stores the result in c17.
sort c17 c13 c14 c18 c19 c20         # c19 c20 contain X and Y for the ray tip.

# -----------------------------------------------------------------------

Plot c1*c2;                          # Plots the coordinates of the sun.
nodt;                                # No default title.
Polygon c19 c20;                     # Draws the rays as a polygon.

  esize k10;                         # The size of the edge of the polygon.

  ecolor k4;                         # The colour of the polygon.

Noframe;                             # Remove frame including tickmarks etc.

Data 0 1 0 1;                        # The data area covers the whole area.

  Etype 0;                           # No frame around the plot.

Symbol;                              # 

  Type 6;                            # Type of symbol.

  Size k1;                           # Size of symbol.

  Color k4;                          # Colour of symbol.
scale 1;                             # The X-axis.
  Max 1;                             # The maximum and minimum commands span

  Min 0;                             # here the are to the range 0 – 1 in both axes.

scale 2;                             # The Y-axis.
  Max 1;                             # 
  Min 0;                             # 
Text k11 k12 'SEMC';                 # Coordinates for the text 'SEMC'.

  Placement 0 0;                     # Places the text centered on the point.

  Tsize 2;                           # Size of 'SEMC'-text.

  Tfont 1.                           # Type of font for 'SEMC'.
Some changes.  Copy all the commands to the 'Command Line Editor' ([Edit]>…) and change the following con​stants to these new values before activating the commands: 
Let k4 = 2                           # Colour of the sun and the rays.

Let k5 = 200                         # Lower end of number of rays.

Let k6 = 500                         # Upper end of number of rays.

 Several graphs on the same page
Using graphs is a very central part of statistical analysis but need to be compiled with care. In order to create a convenient and informative diagram, it must be possible to place several graphs together. In Minitab this is done using the LAYOUT/ENDLAYOUT-feature. The layout-area has its origin (0, 0) in the lower left corner and the upper right corner has the coordinates (1, 1). Placing a single graph in the layout, the X- and Y-distances of the graph edges are specified using the DATA-command showing ”left edge, right edge, bottom edge, top edge”. 
E.g. ”data 0.10 0.30 0.50 0.80;” in the macro means that the graph is to be placed in an area enclosed in a rec​tangle with is left edge X = 0.10, right edge X = 0.30, bottom edge Y = 0.50, top edge Y = 0.80 in the layout-area:
Random 130 c1 c2;               # Stores 130 data in c1 c2 from a exponen-

Exponential 2.1.                # tial distribution, parameter value = 2.1.
A.

Layout;                         # Opens a page for placing graphs and texts.

Title '(A) Several graphs on the same page'.

Histogram c1 c2                 # Places two histograms on the page by the

                                # default placement (data 0.2 0.8 0.2 0.8).

endlayout                       # Closes the page and draws its contents. 
B.

Layout;                         # Opens a page for placing graphs and texts.
Title '(B) Several graphs on the same page'.
Histogram c1;                   # Places the first histogram on the page

Data 0.2 0.4 0.5 0.7.           # using the info of the data-command.
Histogram c1;                   # Places the first histogram on the page

Data 0.6 0.9 0.3 0.5.           # using the info of the data-command.

endlayout                       # Closes the page and draws its contents.
C.

Layout;                         # Opens a page for placing graphs and texts.

Title '(C) Several graphs on the same page'.

Histogram c1;                   # Places the first histogram on the page
Data 0.1 0.4 0.6 0.9;           # using the info of the data-command.
nodt;                           # 'No default title'.
Noframe.

Histogram c2;                   # Places the first histogram on the page
Data 0.6 0.9 0.4 0.7;           # using the info of the data-command.
nodt;                           # 'No default title'.
Noframe.

Plot c1*c2;                     # Plots c1 as ’Y’ and c2 as ’X’ using the
nodt;                           # 'No default title'.
Data 0.08 0.46 0.08 0.47;       # specified data area

Etype 0.                        # with no enclosing rectangle.

endlayout                       # Closes the page and draws its contents.
[image: image1.jpg]

 3D diagrams
Sometimes there is a need to plot 3D-data. Usually this is done in two ways either the very data itself as a scatter diagram in space or as a mathematical function:
Help pltx                            # Gives some info for a 3D-plot.

Random 12 c1-c3                      # Generates 12 values from N(0, 1).

Pltx c1*c2*c3                        # A 3D scatter plot.

Pltx c1*c2*c3;                       # A 3D scatter plot with lines

Symbol;                              # projected to the base.

Project.

Let c4 = 13 + 4 * c2 + 7 * c3        # c4 as a function of c2 and c3.

Pltx c4*c2*c3;                       # A 3D scatter plot with lines

Symbol;                              # projected to the base.

Project.

Mesh c11 c12;                        # Creates a mesh of data in c11 and c12.
 XMesh -3 3 20;                      # Data for the X-direction.

 YMesh -3 3 20.                      # Data for the Y-direction. 

Plot c11 * c12                       # Shows the grid of points.

Let c14 = 13 + 4*c11 + 7*c12         # c4 as a function of c11 and c12.

Pltx c14*c11*c12;                    # A 3D scatter plot with lines

Symbol;                              # projected to the base.

Project.

Surfaceplot c14*c11*c12;             # Creates a surface plot.

   Light 1 0 0 1;                    #

     Color 1.                        #
Alternative 1

[Calc]>[Make Mesh Data…] and enter the various fields. Choose the bivariate normal as a function.

[Graph]>[3D Surface Plot…] choose 'Wireframe' and enter the various fields.

Alternative 2

	Suppose these three functions are to be plotted:

Mesh c11 c12;                        # Creates a mesh of data in c11 and c12.
  XMesh -3 3 20;                     # Data for the X-direction.

  YMesh -3 3 20.                     # Data for the Y-direction. 

Let c13 = sin(c11) + cos(c12)        # First function at 'Alternative 2'.

Let c14 = sin(c11) * cos(c12)        # Second function at 'Alternative 2'.

Let c15 = sin(c11) + cos(c12) + sin(c11) * cos(c12) # Third function at 'Alternative 2'.
	[Graph]>[3D Surface Plot…] use 'Wireframe' and enter the fields, i.e. c13 in ”Z”, c11 in ”X” and c12 in ”Y”.

 Using the HELP-function
When using a software there is a constant need for help about various issues. One such help is after performing a certain analysis. This help is obtained via the ''-button (top row of the screen or by the buttons SHIFT + F1). 

The menu [Help] provides several sub-items to be explored. The quickest way to get help about a certain com​mand is by typing HELP and the command: 

help plot
	[image: image5.png]=lo/x|
@ - o & B

Show ok Fowad  Pint  Opfions

PLOT (high-resolution)

B examples ses aiso
Graph > Scatterplot

Command Syntax

PLOTC *C Displays a scatferpiotof y (1t columm) by x (2nd column).

PLOTC*C..C*C  Displays mulile scatterplts.

Scatterpiots lustratethe refationship between two variables by plting one against the other on the y- an x-axes. You oan also use PLOT with a
datefime varisbleto creste a time series lct, incluing time series plots with regtar intervals (which cannct be handled by TSPLOT).

PLOT wih no subcommnds displays symbols. Lse SYMEOL f you wart symbals to be cisplayed wih CONNECT, PROJECT, AREA, LOWESS, or
REGRESS.

This Syntax.. Generates..
PLOT RevenuetYesr Expenset¥esr Two graphs: Reveue by Year and Expense by Year
PLOT (Revenue Expense]*Year Two graphs: Revenue by Year and Expense by Year
PLOT (Revenue Bxpensel® (Quarter Year) FOUr Graphs: Revenus by Quarter, Revenue by Year, Expense by Guarter, and Expense
by vear
PLOT (Sales Advertisi*Years One graph af Ssles and Adverts by Year wih ifferent symbols representing Sales and
ovERLAY. aavertis

Subcommands.

Seale Labels Data View, Muttiple Graphs  Annotation
axLapEL DaTLAR ARES, OvERLAY. ELLPSE
cRD FoONOTE comnEct PANEL e
MGRD. NODFOOTNOTE ATTER SAME MARKER
NOFRAME NODSUBTITLE Longss SEPARATE PoLYGON
REFERENCE NODTITLE NOJITTER Data Options BECTANGLE
scalE NOPERFOOTNOTE  PROJECT BXCLUDE TEXT
KvSANE NOSEPSUBTITLE REGRESS EREQUENCY. L
Graph Window, suBmILE SvwBoL hcLube
Input, and Output  TTLE NOEMPTY. FloURE
GERINT. NOMISS. LecEnD
GsAvE NOLEGEND
wiitLe cRapH





	This example shows the help available for the command PLOT.

Most of the commands have one or more levels down to obtain the detailed sub-items.

This example shows the PLOT and its sub-item SCALE.

	The SCALE-command, that must be written after the PLOT, con​trols the details of the appea​rance of the axis.

See the example below.


random 20 c1 c2                       # Creates some data.

plot c1*c2;                           # Creates a diagram.

scale 1;                              # The SCALE and LDISPLAY removes the 

Ldisplay 0 0 0 1.                     # X-axis and its labels.

 Commands creating some very simple graphs
The graphical possibilities of the program give endless ways of creating the details of a dia​gram. Below there are a number of diagrams that use only a fraction of these possibilities. However, they can serve as a model in the understanding of how the graphical features are designed. 

 1  Simplest plot command

random 10 c1 c2

plot c1*c2;

Footnote 'Example 1: Simplest plotting command'.

 2  No Difference. 'Figure' fills the whole plot area

random 10 c1 c2

plot c1*c2;

figure 0 1 0 1;

Footnote 'Example 2: No difference'.

 3  The diagram fills the whole plot area. 'Data' as large as 'figure'

random 10 c1 c2

plot c1*c2;

figure 0 1 0 1;

data 0 1 0 1;

Footnote 'Example 3: Fills the whole area'.

 4  Slightly smaller. See example 3

random 10 c1 c2

plot c1*c2;

figure 0.05 0.95 0.05 0.95;

data 0 1 0 1;

Footnote 'Example 4: Slightly smaller. Compare Example 3'.

 5  A very small plot

random 10 c1 c2

plot c1*c2;

figure 0.5 0.6 0.5 0.6;

data 0 1 0 1;

Footnote 'Example 5: Very small plot'.

 6  Filling the whole area using the DATA-command

random 10 c1 c2

plot c1*c2;

data 0 1 0 1;

Footnote 'Example 6: Fills the whole area using "data 0 1 0 1"-command'.

 7  Several histograms in the same diagram – the LAYOUT-command
random 100 c1 c2

layout;

title 'Example with the layout-command';

Footnote 'Example 7: The use of LAYOUT'.

histogram c1;

figure 0.1 0.4 0.1 0.4;

data 0 1 0 1.

histogram c2;

figure 0.5 0.8 0.5 0.8;

data 0.5 0.8 0.5 0.8.

endlayout

 8  'connect' connects the data

random 10 c1 c2

plot c1*c2;

connect;

Footnote 'Example 8: Line between data points'.

 9  Two different diagrams

random 10 c1 c2

plot c1*c2 c2*c1;

connect;

Footnote 'Example 9: Gives two different diagram: c1*c2 and c2*c1'.

 10  Two diagrams in the same plot

random 10 c1 c2

plot c1*c2 c2*c1;

overlay;

connect;

Footnote 'Example 10: Two graphs in the same plot: OVERLAY'.

 11  '+' as a symbol

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;

Footnote 'Example 11: "+" as plotting symbol'.

 12  A large symbol

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;

  size 5;

Footnote 'Example 12: Large symbol'.

 13  Y-values of the plot points

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;

  label 'point 1' 'point 2';

Footnote 'Example 13: Text at the plot points'.

 14  Y-values angled

random 10 c1 c2

plot c1*c2;

datlab;

yvalues;

  angle 45;

Footnote 'Example 14: Other text (angled)'.

 15  Fixed scale in X

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;
scale 1;
min -5;

max  5;

Footnote 'Example 15: Fixed scale in X'.

 16  Fixed scale in X and Y

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;

scale 1;

min -5;

max  5;

scale 2;

min -15;

max  15;

Footnote 'Example 16: Fixed scale in X and Y'.

 17  Without scales and tick marks

random 10 c1 c2

plot c1*c2;

symbol;

  type 2;

noframe;

footnote 'Example 17: No scales or tick marks'.

 18  Without the frame

random 10 c1 c2

plot c1*c2;

scale 1;

hdisplay 0 0 0 0;

scale 2;

hdisplay 0 0 0 0;

footnote 'Example 18: No frame'.

 19  Plot area with equal scales

random 500 c1 c2

plot c1*c2;

connect;

xysame;          # Gives plotting area with the same scales.

footnote 'Example 19: Plot area with the same scales';

tsize 0.8.

 20  Same as 19 but with different plot order

random 500 c1 c2

plot c1*c2;

connect;

order 2;     # Same as before but in different order

xysame;

footnote 'Example 20: Same as Ex 19 but with different plot order';

tsize 0.8.

 21  Plotting in column order

random 500 c1 c2

plot c1*c2;

connect;

  color 2;    # Line with colour

  order 0;    # Order according to column.

footnote 'Example 21: Plotting in order according to column'.

 22  Same as 21 but larger

random 500 c1 c2

plot c1*c2;

title 'Same as Ex 21 but larger';

tsize 1.1;

data 0.07 0.93 0.09 0.90;

connect;

  color 2;

  order 0;

scale 1;

label '';

min  -3;

max  3;

scale 2;

min  -3;

max  3;

axlabel 1 ' ';

footnote 'Example 22: Same as Ex 21 but larger';

tsize 0.8.

 23  Text on the axes

random 10 c1 c2

plot c1*c2;

axlabel 1 'Describes the X-axis';

axlabel 2 'Describes the Y-axis';

Footnote 'Example 23: Text on the axis'.

 24  Same as 23 but with larger text

random 10 c1 c2

plot c1*c2;

Title 'Same as Ex 23 but with larger text (X-axis)';

axlabel 1 'Large text: X-axis';

tsize 3.2;              # Gives larger text

axlabel 2 'Describes the Y-axis';

Footnote 'Example 24'.

 Some advice creating graphs
A diagram is usually a good way to transfer knowledge and information. However, there are several things that need to be considered. The creator must consider the reader and the data itself as the dia​gram must not be an artistic creation with all possible colours, fonts, Clip Arts, etc. Anything that attracts attention without adding value must be avoided. 
Many graphs are routinely created as part of e.g. weekly or quarterly reports. Quite often these graphs need to be redesigned or reevaluated in order to stand up to acceptable standards. 

The variation of the data.  If the data contain variation, this must be visualized. The variation is the main trouble to many problems but also a possibility in finding improvements and solutions.

Improvements, changes of levels/averages, etc.  If the data contain improvements, changes of levels, etc, these must be shown and not only described verbally. 

Stratification of the data.  'Stratification' means that the data is divided into logical parts by e.g. machine, production line, customer, etc. This can be achieved by using e.g. different symbols, different colours, etc (%Stratify).
Use of histograms.  A histogram is a graphical entity that mirrors the true distribution of the data. A histogram should be used when applicable (%Hist).

Use of probability plots.  A probability plot is often very useful when comparing many sets of data (%Hist). 

Use of SPC-diagrams (Statistical Process Control).  If possible draw diagrams that contain a time axis, as a SPC-diagram with calculated control limits. Make sure that the correct diagram is chosen. E.g. measurements of cleanliness of the air (particles per unit of volume) are a Poisson variable and should be plotted as a so-called 'u-diagram'.

If the data is of the type 'number of incorrect items in samples of n items' then a so-called 'p-diagram' should be used. (If n varies between samples then the control limits also will vary).

3D-drawing of a 2D-diagram.  Do not make a 3D-appearance of a two dimensional set of data. Some computer programs can make e.g. bars or pies with a 3D-appearance. This does not add any value to the diagram and should be avoided.

Labels and other annotation on the axes.  Make sure that the labels of the axes are clear, read​able and complete. E.g. if one label states "Diameter" then make it more complete by writing "Diameter (mm)" or what ever is applic​able. Other information regarding the data in the diagram, should also be shown, e.g. time span, number of data points, overall average and standard deviation, etc.

Keep the data area clean.  Do not clutter the data area (the rectangular area that might contain data points) by e.g. arrows, labels, text boxes, legends, etc; keep these simple and to a minimum and enter them outside the data area.

Comments of the diagrams.  Sometimes the diagram contains information that need to be com​mented. These comments should be found close to the graphs. If a graph relies on or refers to some special numerical table this should also be easily available. If data is omitted of some reason or other, this should be made obvious.

Use of lines, etc.  For a diagram showing a time series, i.e. a result over time, some kind of line together with small symbols should be used. Do not use bars for this type of data (a bar should be used when there is a reason to show area, like in a histogram). 

Also, by using a symbol for each point together with a line, a gap in the data can easier be seen.

Use of a so-called matrix diagram.  If the data contains several measurements on the same item, i.e. several physical dimensions, the data should be shown as a so-called matrix diagram. This is a matrix of ordinary XY-diagrams placed on one single area. This makes it easier to see and understand relationships between variables.

Change of scales, transformations.  Sometimes the information in a diagram can be shown more clearly if the scales are transformed; either the Y- or the X-axis or both. Common change of scales is to take the logarithm of the data values.

A page with several graphs.

